


# VIDEO BASIC

20 VIDEOLEZIONI DI BASIC  
PER IMPARARE CON LO SPECTRUM



**GRUPPO  
EDITORIALE  
JACKSON**

*Cos'è un robot*  
*Come insegnare a un robot*  
*Il futuro della robotica*  
*Progetto e codifica  
di un carattere*  
**READ DATA**  
**RESTORE**  
**USR BIN**  
*Assegnazione dati*  
*Videogioco n° 11*

# 11

# Spectrum

16K/48K/PLUS



## VIDEO BASIC SPECTRUM

Pubblicazione quattordicinale  
edita dal Gruppo Editoriale Jackson

### Direttore Responsabile:

Giampietro Zanga

### Direttore e Coordinatore

Editoriale: Roberto Pancaldi

Autore: Softidea - Via Indipendenza 88 - Como

### Redazione software:

Francesco Franceschini, Roberto Rossi,  
Alberto Parodi, Luca Valnegri

### Segretaria di Redazione:

Marta Menegardo

### Progetto grafico:

Studio Nuovaidea - Via Longhi 16 - Milano

### Impaginazione:

Silvana Corbelli

### Illustrazioni:

Cinzia Ferrari, Silvano Scolari

### Fotografie:

Marcello Longhini

### Distribuzione: SODIP

Via Zuretti, 12 - Milano

### Fotocomposizione: Lineacomp S.r.l.

Via Rosellini, 12 - Milano

### Stampa: Grafika '78

Via Trieste, 20 - Pioltello (MI)

### Direzione e Redazione:

Via Rosellini, 12 - 20124 Milano

Tel. 02/6880951/5

Tutti i diritti di riproduzione e pubblicazione di  
disegni, fotografie, testi sono riservati.

© Gruppo Editoriale Jackson 1985.

Autorizzazione alla pubblicazione Tribunale di  
Milano n° 422 del 22-9-1984

Spedizione in abbonamento postale Gruppo II/70  
(autorizzazione della Direzione Provinciale delle  
PPTT di Milano).

Prezzo del fascicolo L. 8.000

Abbonamento comprensivo di 5 raccoglitori L. 165.000

I versamenti vanno indirizzati a: Gruppo  
Editoriale Jackson S.r.l. - Via Rosellini, 12  
20124 Milano, mediante emissione di assegno  
bancario o cartolina vaglia oppure  
utilizzando il c.c.p. n° 11666203.

I numeri arretrati possono essere  
richiesti direttamente all'editore  
inviando L. 10.000 cdu. mediante assegno  
bancario o vaglia postale o francobolli.

Non vengono effettuate spedizioni contrassegno.

---

---

# SOMMARIO

---

---

## HARDWARE ..... 2

Cos'è un robot. Come insegnare  
a un robot. Sviluppatori futuri.

## IL LINGUAGGIO ..... 10

Definizione caratteri.  
USR, BIN, READ, DATA  
RESTORE

## LA PROGRAMMAZIONE ..... 28

Calendario.  
Definizione di un carattere.

## VIDEOESERCIZI ..... 32

---

---

## Introduzione

*In questa lezione vedremo innanzitutto  
come funzionano, come apprendono e  
come lavorano i robot, accennando  
anche alle affascinanti prospettive e  
possibilità che questi dispositivi  
promettono di aprire nella vita di tutti i  
giorni.*

*Scopriremo quindi in quale modo sia  
possibile inserire permanentemente in  
un programma informazioni utili o  
ricorrenti, utilizzando le istruzioni  
READ, DATA e RESTORE.*

*Per finire, vedremo un esempio sulla  
tecnica da adottare per personalizzare  
le visualizzazioni in uscita, imparando  
come definire e memorizzare nuovi  
caratteri all'interno del nostro  
computer.*



**Gruppo Editoriale  
Jackson**

# HARDWARE

## Cos'è un robot


Sin da prima dell'avvento dei calcolatori elettronici, una delle maggiori aspirazioni dell'uomo fu quella di riuscire ad inventare e costruire nuove macchine, che potessero sostituirlo nei lavori più faticosi, pericolosi o ripetitivi.

La storia ce lo insegna: ciascun gradino della scala del progresso è stato (ed è tuttora) normalmente caratterizzato da una grande e fondamentale invenzione o scoperta (come il fuoco, il ferro, il vapore o l'elettricità), che - inizialmente realizzata per risolvere determinati problemi - una volta applicata e sviluppata ci ha procurato (e ci procura) nuovi e magari inaspettati progressi e perfezionamenti.

Il calcolatore elettronico, per quanto non possa essere considerato una invenzione nel senso stretto della parola, non sfugge a questa regola. Nato per "calcolare", il computer è diventato la base per una infinità di nuove applicazioni, che su di esso si basano e dal quale dipendono strettamente. Tra queste la robotica, cioè la scienza che si occupa dell'automatizzazione del lavoro attraverso macchine controllate da computer.

Robot è un termine che nella mente di molte persone suscita tuttora parecchi dubbi e perplessità: il ricordo di uomini metallici e di macchine parlanti, tanto cari alla fantascienza più

commerciale, ne rendono forse difficile l'esatta comprensione. Per prima cosa cerchiamo allora di spiegarne il significato. Un robot non è altro che una macchina automatica, la quale, sotto il controllo di un computer, svolge un lavoro prestabilito: ad esempio verniciare un'automobile, avvitare dei bulloni o saldare delle lamiere. Il grosso vantaggio di un robot rispetto ad una normale macchina automatica è che, essendo collegato ad un computer, può essere programmato - con modifiche di solito non





# HARDWARE

molto dispendiose - per svolgere lavori diversi (non troppo diversi, però!).

Alla programmabilità si aggiunge un altro pregio, grazie ai progressi della microelettronica i robot vengono infatti dotati di particolari dispositivi, chiamati sensori, che - per quanto ancora ben

lontani dai sensi dell'uomo - permettono ad un robot di riconoscere il verificarsi di certe situazioni, particolari od anomale. Facciamo un esempio. Supponiamo di dover avvitare con un bullone due pezzi di metallo appositamente predisposti. Se utilizziamo per questa operazione un robot, possiamo metterci al riparo da ipotetiche evenienze, dovute al cattivo allineamento dei due pezzi, al difettoso fissaggio del bullone od all'errata dimensione di uno qualsiasi dei componenti. Fornendo alla macchina degli appositi sensori e programmando opportunamente il calcolatore di controllo è possibile arrestare la macchina al verificarsi di uno di questi casi. Una macchina normale, in presenza di uno di questi casi anomali,

proseguirebbe imperterrita nel proprio lavoro, provocando magari la messa in produzione di un pezzo difettoso o, peggio ancora, la rottura della macchina stessa. L'introduzione del robot inserisce quindi nella catena di montaggio un elemento di controllo e di verifica, oltre che di produzione. Molti robot possono inoltre svolgere lavori che per le persone sarebbero pericolosi o sgradevoli, come misurare la radioattività in una centrale nucleare, disinnescare una bomba o operare in atmosfere inquinate.




# HARDWARE

## Come insegnare ad un robot

La maggior parte dei robot esistenti lavora ed opera in fabbriche in cui tutto è organizzato intorno a loro: di solito debbono prelevare dei pezzi da nastri trasportatori, lavorarli seguendo una certa successione di

operazioni elementari e depositarli infine su un altro nastro trasportatore, che provvede ad avviarli verso altre lavorazioni. Le prestazioni di un



# HARDWARE

robot dipendono comunque da molti fattori, non ultimo quello economico; esistono quindi (e sono la maggioranza, visto il loro minor costo e la relativa semplicità) robot,

costruiti per un uso abbastanza generico, che di volta in volta vengono adattati per svolgere altre mansioni più specifiche. I robot più comuni sono quelli cosiddetti "a braccio": la loro meccanica è infatti costruita prendendo a modello le braccia dell'uomo. Tale disposizione permette - proprio come per le braccia vere - una notevolissima libertà e capacità di movimenti, consentendo alle "mani" (che normalmente sono pinze comandate da motori elettrici) di operare in posizioni altrimenti inaccessibili. Attualmente si sta anche studiando la maniera di dotare queste "mani" di una sorta di senso del tatto, consentendo così una presa adeguata al peso ed alla solidità dell'oggetto da sollevare (finora nessun robot riesce a distinguere un pezzo di ferro a forma d'uovo da un uovo vero, con risultati facilmente intuibili). Progettare, costruire ed insegnare ad un robot è pertanto un'operazione per niente semplice: occorre innanzitutto analizzare e suddividere l'azione che il robot

dovrà svolgere in tutti i possibili ed immaginabili passaggi elementari (ti ricordi gli algoritmi? ...), classificandone così le diverse necessità di movimento. In base a questi movimenti - ed in funzione dello sforzo necessario per compiere ciascuna azione - si deciderà il numero e la posizione dei bracci meccanici, dei giunti (cioè degli snodi tra i singoli bracci) e dei relativi motori.

Per eseguire il proprio lavoro il robot dovrà inoltre essere provvisto di sensori di vario tipo (fotoelettrici, elettronici o meccanici), appropriati al genere di lavorazione richiesto. Attraverso una o più interfacce questi sensori dovranno quindi inviare le informazioni codificate all'elaboratore di controllo, addetto alla direzione ed al coordinamento dei vari movimenti, che provvederà così a comandare l'esecuzione, la modifica o l'arresto delle operazioni previste in origine.

Il computer di controllo andrà pertanto programmato in modo adeguato a queste necessità, tenendo in debito conto tutte le possibilità di imprevisti

# HARDWARE

od anomalie. E questo per decine o centinaia di singole azioni.

Ancora una volta ti puoi rendere conto di quanto sia importante - prima di passare alla fase vera e propria di programmazione - valutare nei suoi più piccoli particolari tutto il complesso di operazioni, indipendentemente dal fatto che siano importanti o insignificanti che il microprocessore del robot - una volta divenuto operativo - sarà sempre chiamato a controllare.

In un robot, soprattutto se di tipo industriale, questa fase di analisi è di particolare difficoltà, visto che di solito un unico microprocessore non è in grado di affrontare da solo l'intera mole di lavoro che il robot deve poter svolgere. Molto spesso (per non dire sempre) si

rende quindi necessario programmare una serie di microprocessori che sviluppino tutte le operazioni particolari (come, per esempio, muovere un braccio o afferrare un oggetto), coordinandoli attraverso un'ulteriore, grossa unità centrale di controllo, che riesca a seguire ogni momento della fase di lavorazione.

Come ben puoi immaginare, se già è difficile programmare un solo calcolatore, programmarne 10 o 15 (perché possano oltre tutto lavorare in collaborazione tra loro) diventa un'impresa estremamente delicata.

## Sviluppi futuri

La robotica è una scienza molto giovane, si può dire appena agli inizi; nel suo futuro vi potranno quindi essere molti sviluppi e miglioramenti, grazie anche ai numerosi progetti di ricerca attualmente in corso in tutto il mondo.

In questo momento una delle strade più battute (a onor del vero, non solo nella robotica) è quella della cosiddetta

“intelligenza artificiale”. Detta in breve, l'intelligenza artificiale sta cercando di dotare i calcolatori di una sorta di intelligenza autonoma ed automatica, capace di apprendere e ricordare gli eventi passati per applicarli in







# HARDWARE



# HARDWARE



# HARDWARE




quelli futuri (proprio come succede per i bambini che una volta scottati non cercano più di toccare il fuoco). Finora i risultati non sono stati molto incoraggianti, ma sembra che entro pochi anni si potrà cominciare a vedere qualcosa. Comunque, al di là di qualsiasi timore che può incutere l'idea di macchine intelligenti, un robot capace di apprendere dall'ambiente circostante consentirebbe di risolvere molti dei problemi legati alla progettazione ed alla programmazione del robot stesso, evitando quella lunga e complessa fase di studio ed analisi che abbiamo visto essere uno degli scogli più impegnativi nella costruzione e nell'installazione di tutti i dispositivi automatici.

Anche se gli sviluppi della robotica sembrano promettere per il prossimo futuro cose molto interessanti, al momento attuale, permangono ancora diversi problemi: i robot non sanno muoversi come si vorrebbe (nessuno a tutt'oggi è riuscito a far camminare un robot come un uomo), costano abbastanza cari (soprattutto i robot industriali) e non riescono ad essere adattati per un uso universale. In commercio ne esistono comunque già di tantissimi tipi, addirittura anche per uso personale. Lo scopo di questi ultimi, più che altro, è però rivolto all'insegnamento dei rudimenti della robotica e solo raramente consentono di ottenere prestazioni adatte per un'applicazione reale. Essi permettono comunque di apprendere con facilità, e con una spesa relativamente limitata, il funzionamento e la programmazione dei robot, essendo molto spesso collegabili ai più diffusi personal computer per il controllo dei movimenti.

# LINGUAGGIO

## Definizione caratteri



Un carattere è racchiuso in una matrice di 8 punti per 8. Quando vuoi rappresentarlo nei dettagli è necessario operare su una matrice di punti maggiori.

Nell'esempio puoi notare come per riprodurre dettagliatamente un alfabeto particolare si sia fatto ricorso ad una matrice di 16 x 16 (4 caratteri).

# LINGUAGGIO

Accade talvolta che, per necessità o per divertimento, si desideri modificare uno o più caratteri tra quelli normalmente disponibili sul tuo Spectrum. Per fare ciò, è necessario eseguire una semplice operazione, meglio nota come definizione dei caratteri,

**abcdefghijkl**  
**ABCDEFGHIJG**

che consente di creare un nuovo insieme di simboli da far corrispondere a ciascuno dei tasti presenti sulla tastiera. Così, se per esempio volessimo poter disporre sulla tastiera delle lettere dell'alfabeto greco, basterebbe che "insegnassimo" al computer la forma di ciascuno dei nuovi caratteri, in modo che al tasto "A" corrisponda la "alfa", al tasto "B" la "beta", e così via.

Tu sai già cosa fa lo Spectrum per essere in grado di visualizzare tutti i caratteri che puoi di solito vedere sul video, e sai anche che le forme dei vari caratteri (costituite da serie di 8 byte per ogni carattere) si trovano gelosamente custodite nella memoria a sola lettura (ROM), perché non vengano perse ogni volta che spegni il computer.

Quando premi un tasto, questo corrisponde quindi ad un indirizzo della ROM, la quale contiene la sequenza di informazioni che il computer desidera conoscere perché il carattere possa apparire sul video.

Dal momento che la memoria ROM non può

essere alterata in alcun modo, la prima cosa da fare, se si vogliono definire dei nuovi caratteri, è perciò quella di trasferire tutte le immagini dei vecchi caratteri in una zona accessibile anche alla scrittura (cioè nella memoria RAM).

A questo punto tutto è predisposto per accettare le eventuali modifiche che desidereremo effettuare ai vari caratteri. Per imparare a definire un nuovo insieme di caratteri è però necessario esaminare innanzi tutto come si deve fare per creare un singolo carattere. Ciascun carattere è composto da una combinazione di 64 punti (ottenuta


# LINGUAGGIO

utilizzando gli 8 bit di 8 byte), disposti su 8 righe ed 8 colonne. Ecco un esempio, rappresentato da una sequenza di questo tipo:

BINARIO	DEC.	
0 0 1 1 1 1 0 0	60	
0 1 1 0 0 1 1 0	102	
0 1 1 0 1 1 1 0	110	
0 1 1 0 1 1 1 0	110	
0 1 1 0 0 0 0 0	96	
0 1 1 0 0 1 1 0	102	
0 0 1 1 1 1 0 0	60	
0 0 0 0 0 0 0 0	0	

dove ogni 1 o 0 rappresenta rispettivamente un punto (pixel) acceso o spento. Per un computer, acceso o spento significa però valore 1 o 0: basterà allora introdurre nelle locazioni adibite alla definizione di quel carattere gli otto numeri binari, letti per riga, che

definiscono le varie combinazioni di pixel. Nel nostro esempio, prendendo la prima riga (00111100) otteniamo un certo numero binario, che, convertito in decimale (60), ci fornisce uno degli otto valori da inserire nelle locazioni della memoria che specificheranno la




# LINGUAGGIO

descrizione e la struttura di quel carattere.

Per chiarezza questo valore è stato scritto sulla destra della medesima riga.

La stessa operazione sulle altre sette righe ci


procura infine i restanti valori, necessari per definire completamente tutto il carattere.

Gli otto numeri che derivano da  sono pertanto: 60, 102, 110, 110, 96, 102, 60 e 0. A questo punto li si potrà inserire nella memoria. Quando il computer dovrà visualizzare questo carattere preleverà allora gli otto numeri dalla memoria e li visualizzerà sullo schermo. Questo è tutto. Il primo passo da eseguire nella definizione di un carattere è perciò quello di disegnare una serie di 64 quadretti nei quali definire i pixel da accendere o da





spegnere.

Creando un qualsiasi carattere è buona norma cercare di non usare mai uno 0 od un 1 isolato: infatti, adoperando un televisore di qualità non ottima, è probabile che il punto corrispondente non venga visualizzato. Fatta la prima volta, la

Queste videate mostrano le stesse informazioni, rappresentate però con un diverso set di caratteri. La prima utilizza quello normale (presente nella ROM del tuo Spectrum). La seconda ne utilizza uno appositamente definito e memorizzato nella RAM.




# LINGUAGGIO



via da seguire per impostare altri caratteri è sempre la stessa: cambieranno soltanto i numeri da inserire nel computer (ogni carattere è chiaramente definito da una particolare ed unica sequenza di punti

luminosi) e gli indirizzi delle locazioni nelle quali introdurre questi numeri. Alla fine l'insieme dei caratteri sarà stato definito e potrà essere utilizzato a piacimento da qualsiasi programma.





Per recuperare il vecchio insieme di caratteri sarà sufficiente spegnere il computer, riaccendendolo dopo qualche secondo: questa operazione cancellerà tutte le modifiche effettuate nella memoria RAM, ripristinando gli indirizzi della ROM che corrispondono ai soliti tasti.

---

## USR-BIN

---

Il tuo SPECTRUM è in grado di funzionare in vari modi, tra i quali il modo grafico. Cosa significa grafico? Molto semplice: quando ti trovi in modo grafico, (cursore **G**) premendo le cifre dall'1 all'8 appaiono i simboli grafici

rappresentati sui rispettivi tasti, mentre le lettere dell'alfabeto restano inalterate. La cosa, però, non è proprio così banale. La copia dell'alfabeto che si ottiene premendo i tasti in modo grafico non viene infatti letta dallo SPECTRUM nella memoria ROM (come invece accade trovandosi per esempio in modo L), bensì in una zona della memoria RAM, dove viene automaticamente posta al momento dell'accensione del computer. Ciò significa che, volendo, è possibile modificare a proprio piacimento questi caratteri in altri caratteri, definiti e personalizzati

# LINGUAGGIO

dall'utente. Così, alterando opportunamente le locazioni della memoria che contengono per esempio il carattere A, è possibile definire un nuovo carattere, in modo

che da lì in avanti alla pressione del tasto A in modo grafico corrisponda proprio quel carattere.


Abbiamo già visto prima come fare per costruire e definire sulla carta il carattere desiderato.

Una volta trovata la sequenza di 0 e di 1 corrispondente a ciascuna delle 8 righe che identificano il carattere, occorre introdurla nella memoria. Riprendendo l'esempio


del carattere **Q**, avevamo ottenuto:

0	0	1	1	1	1	0	0
0	1	1	0	0	1	1	0
0	1	1	0	1	1	1	0
0	1	1	0	1	1	1	0
0	1	1	0	0	0	0	0
0	1	1	0	0	1	1	0
0	0	1	1	1	1	0	0
0	0	0	0	0	0	0	0

Convertendo questi numeri binari in decimali si ricavano gli otto numeri: 60, 102, 110, 110, 96, 102, 60, 0.



# LINGUAGGIO

Decidiamo che il tasto che dovrà corrispondere al nostro nuovo carattere debba essere la B, in modo tale che quando premeremo B in modo grafico venga visualizzato .

Adesso però sorge un nuovo problema: come trovare gli indirizzi della memoria RAM corrispondenti alla B, nei



quali occorre scrivere questi 8 numeri? La risposta si chiama USR. USR è una funzione che converte un argomento di tipo stringa composto da un solo carattere nell'indirizzo occupato dal primo degli 8 byte del carattere grafico dell'argomento. Quindi:

USR "B"

ci restituirà l'indirizzo nel quale memorizzare il primo degli 8 numeri, USR "B" + 1 il secondo e così via.

Fornendo a USR un argomento di più caratteri si avrà il messaggio di errore INVALID ARGUMENT. Il programma che carica in memoria il carattere grafico da noi definito sarà quindi:

```
10 FOR A = 0 TO 7
20 READ B
30 POKE USR "B" + A, B
40 NEXT A
50 DATA 60, 102, 110, 110
60 DATA 96, 102, 60, 0
```

Una volta eseguito il programma, l'immagine del carattere  sarà stata inserita nel computer e potrà essere richiamata premendo, in modo  (grafico), il carattere B.

Volendo, è possibile saltare la fase di conversione dei numeri da binari in decimali, utilizzando una seconda funzione: BIN.

BIN viene utilizzata per introdurre nello SPECTRUM un numero in forma binaria anziché decimale. Quindi:

POKE 300, BIN 11111111



# LINGUAGGIO

è equivalente a

```
POKE 300, 255
```

infatti

```
PRINT BIN 11111111
```

stampa il risultato 255 decimale.

Il nostro programma avrebbe perciò anche potuto essere scritto così:

```
10 FOR A = 0 TO 7
20 READ B
30 POKE USR "B" + A, B
40 NEXT A
50 DATA BIN 00111100
60 DATA BIN 01100110
70 DATA BIN 01101110
80 DATA BIN 01101110
90 DATA BIN 01100000
100 DATA BIN 01100110
110 DATA BIN 00111100
120 DATA BIN 00000000
```

In pratica la funzione BIN ci permette di introdurre nelle linee DATA la matrice dei punti di un carattere esattamente come l'abbiamo disegnata.

## Sintassi della funzione USR

USR stringa di un solo carattere compreso tra A e U.

## Sintassi della funzione BIN

BIN numero binario di massimo 16 bit

## READ/DATA

Le istruzioni READ e DATA permettono di leggere dei dati all'interno di un programma, evitando così di doverli impostare manualmente dalla tastiera.

L'introduzione dei dati da elaborare avviene inoltre senza il momentaneo arresto del programma, al contrario di quanto accade per l'istruzione INPUT. Con ogni probabilità ti starai però chiedendo quando mai potrà servirti una simile struttura, visto che finora sei riuscito benissimo a farne a meno.

Un esempio ti chiarirà subito le idee.

Accade molto spesso che un programma richieda come azione preliminare dell'esecuzione la cosiddetta inizializzazione delle variabili, cioè l'inserimento di alcuni specifici valori in determinate variabili (per esempio i nomi e le durate dei singoli mesi dell'anno).

Per fare questa operazione si hanno a disposizione tre possibili

# LINGUAGGIO

alternative:

— utilizzare una lunga serie di LET all'inizio del programma;


— richiedere ad ogni esecuzione, utilizzando delle INPUT, i valori da assegnare alle singole

variabili;

— adoperare READ e DATA.

Scartiamo subito la prima soluzione: richiederebbe troppo lavoro durante la battitura del programma

e farebbe occupare troppa memoria (ricorda che ogni istruzione conservata nel computer occupa una certa porzione di memoria). La seconda soluzione è già più accettabile. Nel



# LINGUAGGIO


caso dei mesi si  
potrebbe scrivere:

```
10 DIM M$(12, 10) : DIM G(12)
20 FOR I = 1 TO 12
30 INPUT M$(I), G(I)
40 NEXT I
```

e con queste poche  
istruzioni ce la saremmo  
cavata.

Il problema è però risolto  
solo parzialmente: ad

ogni RUN dovremmo  
infatti metterci di buona  
lena a battere GENNAIO,  
31, FEBBRAIO, 28, ...  
ecc., impostando cioè  
dei valori che, tutto  
sommato, tra  
un'esecuzione e l'altra  
non subiscono alcuna  
modifica e che quindi  
sarebbe comodo  
conservare in  
permanenza nel




# LINGUAGGIO

programma. Ciò significa che ogni volta che utilizziamo questo programma dobbiamo assegnare da tastiera

tutte le 24 informazioni che ci vengono richieste. Lavoro lungo, noioso ed inutile.

Adottando READ e DATA avremo invece:

```
10 DIM M$(12, 10) : DIM G (12)
20 FOR I = 1 TO 12
30 READ M$( I) : READ G (I)
40 NEXT I
50 DATA GENNAIO, 31, FEBBRAIO, 28, MARZO, 31, APRILE, 30
60 DATA MAGGIO, 31, GIUGNO, 30, LUGLIO, 31, AGOSTO, 31
70 DATA SETTEMBRE, 30, OTTOBRE, 31, NOVEMBRE, 30,
   DICEMBRE, 31
```



In pratica queste istruzioni ordinano al calcolatore di fare le stesse cose delle linee viste prima, con la sola differenza che le singole variabili degli array M\$ ( ) e G ( ) devono essere lette (READ) non più dalla tastiera - come accadeva prima a causa dell'INPUT - ma dalle righe DATA. I dati vengono quindi inseriti nel calcolatore una sola volta e, cosa ancora più importante, chi usa il programma non è tenuto a conoscere ed a battere questi dati (prova a pensare ai nomi di tutte le squadre del campionato di calcio o ad altre serie di dati, magari meno noti dei nomi dei mesi dell'anno).

# LINGUAGGIO

In questo modo è possibile mantenere - conservati all'interno del programma - dei valori che altrimenti andrebbero tutte le volte dispersi con lo

spegnimento del calcolatore o con la riesecuzione del programma stesso. Le istruzioni DATA contengono i diversi valori che si vogliono assegnare. Esse permettono di riunire tutti i dati in un unico punto del programma, di solito all'inizio o alla fine, dove sono più leggibili. Ogni termine di tipo stringa deve, come al solito, essere racchiuso entro virgolette. L'istruzione READ consente di leggere i dati che sono memorizzati nelle DATA. È ovvio che si debba fare estrema attenzione affinché il tipo della variabile dell'istruzione READ corrisponda al

contenuto dell'istruzione DATA, proprio come succede negli INPUT da tastiera: non si dovrà mai verificare che una variabile numerica possa contenere un valore alfanumerico o che in una variabile di tipo stringa si trovi un valore numerico. Questo tipo di errore è sempre in agguato!

Le linee DATA del listato precedente sono tre solo per motivi di leggibilità: avremmo infatti potuto metterne un numero diverso anche non necessariamente consecutive. Infatti se vi sono più linee DATA, in punti diversi del programma, queste sono lette di seguito, come se si trattasse di un'unica linea.

## Esempi

```
10 READ A: READ B
20 DATA 3, 5
```

Alle variabili A e B vengono assegnati i valori 3 e 5.

```
10 READ A
20 READ B: READ C
30 DATA 4, 7
```

Non ci siamo: alla variabile C non corrisponde alcun valore e si avrà il messaggio di errore OUT OF DATA.



# LINGUAGGIO

```
10 READ A
20 READ A$
30 READ C
40 DATA 3, "4", 5
```

A, A\$ e C assumono rispettivamente valore 3, "4" e 5.

Nota le virgolette sul 4: la seconda variabile è infatti di tipo stringa. Il 4 verrà quindi inserito in memoria come carattere e non come numero.

---

```
10 READ C$, A , D$
20 DATA "PIPPO", "PLUTO"
```

In questo caso gli errori sono due: la variabile A, essendo di tipo numerico, non può assumere valore PLUTO (apparirà il messaggio di errore NONSENSE IN BASIC) e la variabile D\$ è sprovvista del corrispondente DATA.

---

```
10 FOR I = 1 TO 6
20 READ K$
30 PRINT K$
40 NEXT I
50 DATA "FRANCO", "MARIO"
60 DATA "MATTEO", "CARLO"
70 DATA "PIPPO", "GIACOMO"
```

La variabile K\$ assumerà via via i 5 valori specificati nelle istruzioni DATA. Il comando PRINT K\$, inserito nel ciclo FOR, provocherà pertanto la stampa di FRANCO, MARIO, ..., GIACOMO. Nota come la variabile K\$ sia dello stesso tipo (stringa) dei valori che dovrà assumere in seguito alle READ.

# LINGUAGGIO

```
5 CLS
10 PRINT "SISTEMA SOLARE" : PRINT
15 FOR K = 1 TO 9
20 PRINT
25 READ P$, D
30 PRINT "IL PIANETA "; P$
35 PRINT "HA DIAMETRO DI"; D; "KM"
40 NEXT K
45 DATA "MERCURIO", 4880, "VENERE", 12096
50 DATA "TERRA", 12740, "MARTE", 6780
55 DATA "GIOVE", 141560, "SATURNO", 120800
60 DATA "URANO", 51000, "NETTUNO", 49300
65 DATA "PLUTONE", 7000
```

Questo, più che un esempio, è una dimostrazione dell'utilità di READ ... DATA. Con poche istruzioni è infatti possibile inserire in permanenza delle informazioni che non cambieranno mai tra un'esecuzione e l'altra, consentendo un considerevole risparmio sia di spazio che di tempo.

## Sintassi dell'istruzione DATA

---

DATA costante [ , costante ] [ , ... ]

---

## Sintassi dell'istruzione READ

---

READ variabile [ , variabile ] [ , .... ]

---

## RESTORE

Per controllare l'ordine secondo cui vengono prelevati i valori delle DATA possiamo immaginare che all'interno della memoria esista una specie di indice (chiamato anche puntatore) che segnali - di volta in volta - qual è il successivo valore che può essere letto con un'istruzione READ. Ogni volta che il computer legge un


qualsiasi elemento dalle linee DATA questo puntatore viene perciò fatto spostare sull'elemento successivo, in modo che l'interprete BASIC sappia sempre sin dove è



# LINGUAGGIO

arrivata la lettura dei vari termini delle DATA. Ad ogni READ corrisponderà quindi un automatico avanzamento del puntatore delle DATA.

Talvolta può comunque essere molto utile avere la possibilità, nel corso del programma, di accedere alle stesse informazioni più di una volta.



# LINGUAGGIO

L'istruzione RESTORE serve appunto per riportare il puntatore interno alla prima linea DATA del programma rendendo nuovamente disponibili le costanti

come se non fossero mai state lette. Quando viene incontrata, l'istruzione RESTORE - indipendentemente dal numero di elementi che sono già stati letti - fa quindi in modo che la successiva istruzione READ torni indietro a leggere il primo elemento nella lista dei DATA, proprio come accade la prima volta. Così, il seguente programma

```
10 FOR I = 1 TO 1000
20 READ A
25 PRINT A
30 RESTORE
40 NEXT I
50 DATA 27, 10, 60
```

avrà come unico risultato la stampa per

1000 volte del valore 27, visto che il RESTORE posto alla linea 20 impedisce al puntatore delle DATA di avanzare al termine successivo. Prova adesso a togliere la riga 30 ed a eseguire il programma: se avevi dei dubbi sul funzionamento di RESTORE, li chiarirai subito.

Se lo si desidera, RESTORE può anche essere fatto seguire da un numero di linea. In questo caso il puntatore viene fatto spostare - anziché al primo elemento della prima DATA - al primo elemento della DATA corrispondente al numero di linea specificato.

## Sintassi dell'istruzione

---

RESTORE [ numero di linea ]

---

# PROGRAMMAZIONE

## Calendario

Il primo esempio della nostra lezione è un'utile applicazione sull'uso delle istruzioni READ/DATA. Come risultato dell'esecuzione di questo programma si ha infatti la stampa del calendario di un qualsiasi anno a partire dal 1985. Vediamo in che modo.

Per prima cosa occorre conoscere quale giorno della settimana corrisponde al primo giorno dell'anno prescelto. Ciò può essere fatto in modo molto semplice, tenendo cioè presente una data particolare: per esempio, il 1 gennaio 1984, che era una Domenica.



# PROGRAMMAZIONE

A questo punto è sufficiente calcolare il numero di giorni che intercorrono tra l'inizio dell'anno prescelto e l'inizio del 1984 (tenendo naturalmente conto degli anni bisestili) per conoscere automaticamente in quale giorno della settimana cade il 1° gennaio dell'anno prescelto.

Prendiamo, per esempio, il 1986. Tra il 1° gennaio 1984 ed il 1° gennaio 1986 vi sono 731 giorni (365 + 366 perché l'84 era bisestile).


Adesso basta quindi dividere 731 per 7

$731 : 7 = 104$   
con avanzo di 3

per sapere che il 1° gennaio '86 è un mercoledì. Il 3 di avanzo ci dice infatti che l'inizio dell'86 è 3 giorni dopo l'inizio dell'84: quindi (con scrittura forse non molto corretta)

DOMENICA + 3  
= MERCOLEDÌ'.

A questo punto la strada è tutta in discesa: basterà stampare uno dopo l'altro i vari mesi dell'anno, curando che l'impaginazione sia corretta, per ottenere il nostro calendario.



# PROGRAMMAZIONE

```
10 RESTORE
20 DIM M (12) : DIM M$ (12, 10) : DIM G$ (7, 1)
30 FOR I = 1 TO 7 : READ G$ (I) : NEXT I
40 FOR I = 1 TO 12 : READ M (I), M$ (I) : NEXT I
50 REM
60 REM CALCOLA IL PRIMO GIORNO DELL'ANNO
70 INPUT "ANNO "; AN
80 INPUT "1-VIDEO/2-STAMPANTE"; LINE Z$
90 IF CODE Z$ <49 OR CODE Z$ > 50 OR Z$ = " " THEN GO TO 80
100 IF Z$ = "2" THEN OPEN # 2, "P"
110 IF Z$ = "1" THEN OPEN # 2, "S"
120 LET NG = 0
130 IF AN < 1985 THEN GO TO 70
140 CLS
150 REM TROVA GIORNI
160 FOR I = 1984 + 1 TO AN
170 LET NG = NG + 365
180 IF (I/4 = INT (I/4)) AND (I <> AN) THEN LET NG = NG + 1:
    REM ANNO BISESTILE
190 NEXT I
200 REM TROVA GIORNO SETTIMANA
210 LET NG = NG - (INT (NG/7)) * 7
220 LET J = NG + 1
230 REM
240 IF AN/4 = INT (AN/4) THEN LET M (2) = 29
250 FOR H = 1 TO 12 STEP 2
260 FOR I = H TO H + 1
270 PRINT TAB 4; M$ (I); " "; AN
280 PRINT : PRINT " ";
290 FOR K = 1 TO 7
300 PRINT G$(K); " ";
310 NEXT K
320 PRINT
330 IF J = 1 THEN LET K = 1 : GO TO 370
340 FOR K = 1 TO J - 1
350 PRINT " ";
360 NEXT K
370 LET J = K
380 FOR K = 1 TO M (I)
390 PRINT "0" AND K < 10; K; " ";
400 LET J = J + 1
410 IF J = 8 THEN LET J = 1 : PRINT
420 NEXT K
430 PRINT ""
440 NEXT I
```



# PROGRAMMAZIONE

```
450 IF Z$ = "1" THEN PRINT # 0; AT 1, 8; "PREMI UN TASTO" :  
    PAUSE 0 : CLS  
460 NEXT H  
470 CLOSE # 2 : STOP  
480 DATA "LUNEDI", "MARTEDI", "MERCOLEDI", "GIOVEDI",  
    "VENERDI", "SABATO", "DOMENICA"  
490 DATA 31, "GENNAIO", 28, "FEBBRAIO", 31, "MARZO", 30, "APRILE",  
    31, "MAGGIO", 30, "GIUGNO"  
500 DATA 31, "LUGLIO", 31, "AGOSTO", 30, "SETTEMBRE", 31,  
    "OTTOBRE", 30, "NOVEMBRE", 31, "DICEMBRE"
```

## Definizione di un carattere

Il programma che segue ti permette di definire un carattere grafico e di memorizzarlo in una delle lettere (tra A e U) riservate agli U.D.G. Per disegnare un carattere particolare, metti in pratica le tecniche illustrate in precedenza poi inserisci alla linea 60 del listato i valori decimali corrispondenti.

```
10 INPUT C$  
20 FOR I = 0 TO 7  
30 READ N  
40 POKE USR C$ + I, N  
50 NEXT I  
60 DATA 60, 66, 129, 129, 129, 129, 66, 60
```

# VIDEOESERCIZI

Quali parole (costante stringa) verranno stampate?

```
10 READ A : READ B : RESTORE A * B READ A$
20 PRINT A$
30 STOP
90 DATA 10, 13
100 DATA "VIDEO"
110 DATA "BASIC"
120 DATA "SOFTIDEA"
130 DATA "JACKSON"
140 DATA "SINCLAIR"
150 DATA "SPECTRUM"
160 DATA "16 K"
170 DATA "48 K"
180 DATA "PLUS"
190 DATA "QUANTUM LEAP"
```

Molti discorsi di uomini politici sembrano avere la struttura impostata da questo programma cambiando opportunamente le linee data con vocabili come "congiuntura", "piattaforma" ecc.. ne otterrai degli esempi.

```
10 RESTORE
20 FOR I = 1 TO 2
40 LET A = INT (RND * 10)
50 FOR K = 1 TO A
60 READ Z$ : NEXT K
70 IF I = 1 THEN READ A$
80 IF I = 2 THEN READ B$
90 RESTORE 150 : NEXT I
100 PRINT A$; " "; B$
110 PRINT # 0; AT 1, 0; "ANCORA (S/N)" : LET Z$ = INKEY$ : IF Z$ = " " THEN
GO TO 110
120 IF Z$ = "S" THEN RUN
130 STOP
140 DATA "LA GIRAFFA", "L'UOMO", "LA MUCCA", "L'ASINO", "IL CAVALLO",
"L'ELEFANTE", "IL CANE", "IL GATTO", "LA GALLINA", "IL CONIGLIO"
150 DATA "SGHIGNAZZA", "SCOPPIA", "SALTICCHIA", "SVOLAZZA",
"RIMBALZA", FUGGE, "NITRISCE", "BARRISCE", "MUGGISCE", "SORRIDE"
```





**GRUPPO  
EDITORIALE  
JACKSON**